// Vulnerability label: clean
pragma solidity ^0.4.18;

library SafeMath {

  /**
  * @dev Multiplies two numbers, throws on overflow.
  */
  function mul(uint256 a, uint256 b) internal pure returns (uint256) {
    if (a == 0) {
      return 0;
    }
    uint256 c = a * b;
    assert(c / a == b);
    return c;
  }

  /**
  * @dev Integer division of two numbers, truncating the quotient.
  */
  function div(uint256 a, uint256 b) internal pure returns (uint256) {
    // assert(b > 0); // Solidity automatically throws when dividing by 0
    uint256 c = a / b;
    // assert(a == b * c + a % b); // There is no case in which this doesn't hold
    return c;
  }

  /**
  * @dev Substracts two numbers, throws on overflow (i.e. if subtrahend is greater than minuend).
  */
  function sub(uint256 a, uint256 b) internal pure returns (uint256) {
    assert(b <= a);
    return a - b;
  }

  /**
  * @dev Adds two numbers, throws on overflow.
  */
  function add(uint256 a, uint256 b) internal pure returns (uint256) {
    uint256 c = a + b;
    assert(c >= a);
    return c;
  }
}

contract ERC20Basic {
  function totalSupply() public view returns (uint256);
  function balanceOf(address who) public view returns (uint256);
  function transfer(address to, uint256 value) public returns (bool);
  event Transfer(address indexed from, address indexed to, uint256 value);
}

contract ERC20 is ERC20Basic {
  function allowance(address owner, address spender) public view returns (uint256);
  function transferFrom(address from, address to, uint256 value) public returns (bool);
  function approve(address spender, uint256 value) public returns (bool);
  event Approval(address indexed owner, address indexed spender, uint256 value);
}

contract Ownable {
  address public owner;


  event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);


  /**
   * @dev The Ownable constructor sets the original `owner` of the contract to the sender
   * account.
   */
  function Ownable() public {
    owner = msg.sender;
  }

  /**
   * @dev Throws if called by any account other than the owner.
   */
  modifier onlyOwner() {
    require(msg.sender == owner);
    _;
  }

  /**
   * @dev Allows the current owner to transfer control of the contract to a newOwner.
   * @param newOwner The address to transfer ownership to.
   */
  function transferOwnership(address newOwner) public onlyOwner {
    require(newOwner != address(0));
    OwnershipTransferred(owner, newOwner);
    owner = newOwner;
  }

}

contract BasicToken is ERC20Basic {
  using SafeMath for uint256;

  mapping(address => uint256) balances;

  uint256 totalSupply_;

  /**
  * @dev total number of tokens in existence
  */
  function totalSupply() public view returns (uint256) {
    return totalSupply_;
  }

  /**
  * @dev transfer token for a specified address
  * @param _to The address to transfer to.
  * @param _value The amount to be transferred.
  */
  function transfer(address _to, uint256 _value) public returns (bool) {
    require(_to != address(0));
    require(_value <= balances[msg.sender]);

    // SafeMath.sub will throw if there is not enough balance.
    balances[msg.sender] = balances[msg.sender].sub(_value);
    balances[_to] = balances[_to].add(_value);
    Transfer(msg.sender, _to, _value);
    return true;
  }

  /**
  * @dev Gets the balance of the specified address.
  * @param _owner The address to query the the balance of.
  * @return An uint256 representing the amount owned by the passed address.
  */
  function balanceOf(address _owner) public view returns (uint256 balance) {
    return balances[_owner];
  }

}

contract StandardToken is ERC20, BasicToken {

  mapping (address => mapping (address => uint256)) internal allowed;


  /**
   * @dev Transfer tokens from one address to another
   * @param _from address The address which you want to send tokens from
   * @param _to address The address which you want to transfer to
   * @param _value uint256 the amount of tokens to be transferred
   */
  function transferFrom(address _from, address _to, uint256 _value) public returns (bool) {
    require(_to != address(0));
    require(_value <= balances[_from]);
    require(_value <= allowed[_from][msg.sender]);

    balances[_from] = balances[_from].sub(_value);
    balances[_to] = balances[_to].add(_value);
    allowed[_from][msg.sender] = allowed[_from][msg.sender].sub(_value);
    Transfer(_from, _to, _value);
    return true;
  }

  /**
   * @dev Approve the passed address to spend the specified amount of tokens on behalf of msg.sender.
   *
   * Beware that changing an allowance with this method brings the risk that someone may use both the old
   * and the new allowance by unfortunate transaction ordering. One possible solution to mitigate this
   * race condition is to first reduce the spender's allowance to 0 and set the desired value afterwards:
   * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
   * @param _spender The address which will spend the funds.
   * @param _value The amount of tokens to be spent.
   */
  function approve(address _spender, uint256 _value) public returns (bool) {
    allowed[msg.sender][_spender] = _value;
    Approval(msg.sender, _spender, _value);
    return true;
  }

  /**
   * @dev Function to check the amount of tokens that an owner allowed to a spender.
   * @param _owner address The address which owns the funds.
   * @param _spender address The address which will spend the funds.
   * @return A uint256 specifying the amount of tokens still available for the spender.
   */
  function allowance(address _owner, address _spender) public view returns (uint256) {
    return allowed[_owner][_spender];
  }

  /**
   * @dev Increase the amount of tokens that an owner allowed to a spender.
   *
   * approve should be called when allowed[_spender] == 0. To increment
   * allowed value is better to use this function to avoid 2 calls (and wait until
   * the first transaction is mined)
   * From MonolithDAO Token.sol
   * @param _spender The address which will spend the funds.
   * @param _addedValue The amount of tokens to increase the allowance by.
   */
  function increaseApproval(address _spender, uint _addedValue) public returns (bool) {
    allowed[msg.sender][_spender] = allowed[msg.sender][_spender].add(_addedValue);
    Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

  /**
   * @dev Decrease the amount of tokens that an owner allowed to a spender.
   *
   * approve should be called when allowed[_spender] == 0. To decrement
   * allowed value is better to use this function to avoid 2 calls (and wait until
   * the first transaction is mined)
   * From MonolithDAO Token.sol
   * @param _spender The address which will spend the funds.
   * @param _subtractedValue The amount of tokens to decrease the allowance by.
   */
  function decreaseApproval(address _spender, uint _subtractedValue) public returns (bool) {
    uint oldValue = allowed[msg.sender][_spender];
    if (_subtractedValue > oldValue) {
      allowed[msg.sender][_spender] = 0;
    } else {
      allowed[msg.sender][_spender] = oldValue.sub(_subtractedValue);
    }
    Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

}

contract ThemisToken is Ownable, StandardToken {

  string public constant name = "Themis Token";
  string public constant symbol = "THM";
  uint8 public constant decimals = 18;

  uint256 public angelSupply;                            // Angels sale supply
  uint256 public earlyBirdsSupply;                       // Early birds supply
  uint256 public foundationSupply;                       // Foundation/Community supply
  uint256 public teamSupply;                             // Team supply
  uint256 public marketingSupply;                        // Marketing & strategic supply
  uint256 public optionSupply;                           // Option supply

  address public angelAddress;                           // Angel address
  address public earlyBirdsAddress;                      // Early Birds address
  address public teamAddress;                            // Team address
  address public foundationAddress;                      // Foundation address
  address public marketingAddress;                       // Marketing address
  address public optionAddress;                          // Option address

  /**
   * @dev Constructor that gives msg.sender all of existing tokens.
   */
  function ThemisToken() public {
    totalSupply_        =  3000000000 * 1e18;
    angelSupply         =   300000000 * 1e18;             //  10% -  for private angels sale
    earlyBirdsSupply    =   600000000 * 1e18;             //  20% -  for early-bird sale
    teamSupply          =   450000000 * 1e18;             //  15% -  for team
    foundationSupply    =   750000000 * 1e18;             //  25% -  for foundation/incentivising efforts
    marketingSupply     =   600000000 * 1e18;             //  20% -  for covering marketing and strategic expenses
    optionSupply        =   300000000 * 1e18;             //  10% -  for option

    angelAddress         = 0xD58aE13Eb1e8CDb92088709A6868d32C993FAd74;  // Angel address
    earlyBirdsAddress    = 0xac9EaB9C4c403441fb8592529A3cE534E68246ED;  // Early Birds address
    teamAddress          = 0x4d06220df2BC77C3E47b72611AA79915611Ed23B;  // Team address
    foundationAddress    = 0xDa2AB3712A490cC6Df661E3ae398BeA24434349F;  // Foundation/Community address
    marketingAddress     = 0x9f160Ed0F3B8d5180Eb5cC97c43CF7Fe1efFE02C;  // Marketing/Strategic address
    optionAddress        = 0x47343c223F3605aC541a8eC61e5Fd41EBDdEc9d1;  // Option address

    releaseAngelTokens();
    releaseEarlyBirdsTokens();
    releaseFoundationTokens();
    releaseTeamTokens();
    releaseMarketingTokens();
    releaseOptionTokens();

    //balances[msg.sender] = totalSupply_;
    //Transfer(0x0, msg.sender, totalSupply_);
  }

  // -------------------------------------------------
  // Releases angel supply
  // -------------------------------------------------
  function releaseAngelTokens() internal returns(bool success) {
      require(angelSupply > 0);
      balances[angelAddress] = angelSupply;
      Transfer(0x0, angelAddress, angelSupply);
      angelSupply = 0;
      return true;
  }

  // -------------------------------------------------
  // Releases earlybirds supply
  // -------------------------------------------------
  function releaseEarlyBirdsTokens() internal returns(bool success) {
      require(earlyBirdsSupply > 0);
      balances[earlyBirdsAddress] = earlyBirdsSupply;
      Transfer(0x0, earlyBirdsAddress, earlyBirdsSupply);
      earlyBirdsSupply = 0;
      return true;
  }

  // -------------------------------------------------
  // Releases team supply
  // -------------------------------------------------
  function releaseTeamTokens() internal returns(bool success) {
    require(teamSupply > 0);
    balances[teamAddress] = teamSupply;
    Transfer(0x0, teamAddress, teamSupply);
    teamSupply = 0;
    return true;
  }

  // -------------------------------------------------
  // Releases foundation supply
  // -------------------------------------------------
  function releaseFoundationTokens() internal returns(bool success) {
    require(foundationSupply > 0);
    balances[foundationAddress] = foundationSupply;
    Transfer(0x0, foundationAddress, foundationSupply);
    foundationSupply = 0;
    return true;
  }

  // -------------------------------------------------
  // Releases Marketing supply
  // -------------------------------------------------
  function releaseMarketingTokens() internal returns(bool success) {
    require(marketingSupply > 0);
    balances[marketingAddress] = marketingSupply;
    Transfer(0x0, marketingAddress, marketingSupply);
    marketingSupply = 0;
    return true;
  }

  // -------------------------------------------------
  // Releases Option supply
  // -------------------------------------------------
  function releaseOptionTokens() internal returns(bool success) {
      require(optionSupply > 0);
      balances[optionAddress] = optionSupply;
      Transfer(0x0, optionAddress, optionSupply);
      optionSupply = 0;
      return true;
  }

}